



# Power Apps Patch Function Examples For SharePoint

MATTHEW DEVANEY

LAST UPDATED: 2022-11-05



# Table Of Contents

<b>Introduction</b> .....	<b>4</b>
<b>PATCH A Single Line Text Column</b> .....	<b>5</b>
Scenario #1: Text Input .....	6
Scenario #2: Dropdown.....	6
<b>PATCH A Multiple Line Text Column</b> .....	<b>8</b>
Scenario #1: Text Input .....	10
<b>PATCH A Number Column</b> .....	<b>11</b>
Scenario #1: Text Input .....	12
Scenario #2: Dropdown.....	12
Scenario #3: Slider.....	13
<b>PATCH A Yes/No Column</b> .....	<b>14</b>
Scenario #1: Toggle .....	15
Scenario #2: Dropdown.....	15
<b>PATCH A Person Column</b> .....	<b>17</b>
Scenario #1: ComboBox .....	18
<b>PATCH A Person Column With Multiple People</b> .....	<b>19</b>
Scenario #1: ComboBox .....	20
<b>PATCH A Date Column</b> .....	<b>22</b>
Scenario #1: Date Picker .....	23
<b>PATCH A Date &amp; Time Column</b> .....	<b>24</b>
Scenario #1: Date & Time Picker.....	25
<b>PATCH A Choices Column</b> .....	<b>26</b>
Scenario #1: Combo Box .....	27
Scenario #2: Dropdown.....	27
Scenario #3: Dropdown.....	28

**PATCH A Choices Column With Multiple Values ..... 29**

    Scenario #1: ComboBox ..... 30

**PATCH A LookUp Column ..... 31**

    Scenario #1: Combo Box ..... 32

**PATCH An Image Column ..... 34**

    Scenario #1: Add Picture ..... 35

    Scenario #2: Pen Input ..... 35

**PATCH A Hyperlink Column ..... 37**

    Scenario #1: Text Input ..... 38

**PATCH A Currency Column ..... 39**

    Scenario #1: Text Input ..... 40

**PATCH A Task Outcome Column ..... 41**

    Scenario #1: Combo Box ..... 42

    Scenario #2: Dropdown..... 42

    Scenario #3: Dropdown..... 43

# Introduction

Welcome to my mini-guide on using the Power Apps Patch Function with SharePoint.

Inside you will find examples of how to patch every SharePoint column type. Each page contains *code* and *visuals* to show what the code is doing. **COPY + PASTE** code from the guide into your own app. Just change the datasource name and column names to match your own.

I hope you enjoy this free eBook.



# PATCH A Single Line Text Column

## Goal

Change the Address column (Single Line Text Type) from "123 Fake Street" to "456 Ocean Avenue"

## Input

Customer Locations (SharePoint List)

ID	Customer Name	Address
1	XYZ Widgets Company	123 Fake Street

## Patch Function Code

```
Patch(  
  'Customer Locations',  
  Lookup('Customer Locations', ID=1),  
  {Address: "456 Ocean Avenue"}  
)
```

## Output

Customer Locations (SharePoint List)

ID	Customer Name	Address
1	XYZ Widgets Company	456 Ocean Avenue

## Scenario #1: Text Input

Patch a value found in a *text input* to a Single Line Text field.

Address

### Patch Function Code

```
Patch(  
  'Customer Locations',  
  LookUp('Customer Locations', ID=1),  
  {Address: TextInput_Address.Text}  
)
```

## Scenario #2: Dropdown

Patch a value found in *dropdown* to a Single Line Text field.

Address

456 Ocean Avenue	▼
123 Fake Street	
456 Ocean Avenue	
789 Riverton Lane	

## *Patch Function Code*

```
Patch(  
  'Customer Locations',  
  LookUp('Customer Locations', ID=1),  
  {Address: Dropdown_Address.Selected.Value}  
)
```

# PATCH A Multiple Line Text Column

## Goal

Change the Notes column (Multiple Line Text Type) from *blank* to some *ipsum lorem* text.

## Input

Work Orders (SharePoint List)

ID	Customer Name	Notes
1	XYZ Widgets Company	

## Patch Function Code

```
Patch(
  'Work Orders',
  Lookup('Work Orders', ID=1),
  {
    Notes: "Lorem ipsum dolor sit amet, consectetur adipiscing elit. Morbi
scelerisque varius nisi ac ultricies. Vestibulum eu erat viverra, imperdiet quam vitae,
fringilla elit. Sed dolor ipsum, lacinia id sapien in, fermentum porttitor felis.
Aliquam erat volutpat.

Curabitur ut felis vel risus malesuada hendrerit id in ante. Donec tortor libero, congue
et consequat sit amet, condimentum vitae sapien. Suspendisse venenatis lectus in sapien
mattis eleifend nec id ante. Ut nunc est, viverra nec ullamcorper eget, tempus vel
massa. Proin at diam quis augue fringilla mollis quis vel lorem.

Ut ante nulla, dapibus auctor consectetur in, commodo commodo enim. Cras sollicitudin
elit in maximus ultrices. Nullam lacus nulla, vestibulum sit amet ipsum in, bibendum
euismod eros. Pellentesque tristique bibendum libero sit amet lacinia. Aliquam sagittis
eget risus cursus lobortis.
"
  }
)
```



## Output

Work Orders (SharePoint List)

ID	Customer Name	Notes
1	XYZ Widgets Company	<p> Lorem ipsum dolor sit amet, consectetur adipiscing elit. Morbi scelerisque varius nisi ac ultricies. Vestibulum eu erat viverra, imperdiet quam vitae, fringilla elit. Sed dolor ipsum, lacinia id sapien in, fermentum porttitor felis. Aliquam erat volutpat.</p> <p> Curabitur ut felis vel risus malesuada hendrerit id in ante. Donec tortor libero, congue et consequat sit amet, condimentum vitae sapien. Suspendisse venenatis lectus in sapien mattis eleifend nec id ante. Ut nunc est, viverra nec ullamcorper eget, tempus vel massa. Proin at diam quis augue fringilla mollis quis vel lorem.</p> <p> Ut ante nulla, dapibus auctor consectetur in, commodo commodo enim. Cras sollicitudin elit in maximus ultrices. Nullam lacus nulla, vestibulum sit amet ipsum in, bibendum euismod eros. Pellentesque tristique bibendum libero sit amet lacinia. Aliquam sagittis eget risus cursus lobortis.</p>

## Scenario #1: Text Input

Patch a value found in a *text input* to a Multiple Line Text field.

Notes

Lorem ipsum dolor sit amet, consectetur adipiscing elit. Morbi scelerisque varius nisi ac ultricies. Vestibulum eu erat viverra, imperdiet quam vitae, fringilla elit. Sed dolor ipsum, lacinia id sapien in, fermentum porttitor felis. Aliquam erat volutpat.

Curabitur ut felis vel risus malesuada hendrerit id in ante. Donec tortor libero, congue et consequat sit amet, condimentum vitae sapien. Suspendisse venenatis lectus in sapien mattis eleifend nec id ante. Ut nunc est, viverra nec ullamcorper eget, tempus vel massa. Proin at diam quis augue fringilla mollis quis vel lorem.

Ut ante nulla, dapibus auctor consectetur in, commodo commodo enim. Cras sollicitudin elit in maximus ultrices. Nullam lacus nulla, vestibulum sit amet ipsum in, bibendum euismod eros. Pellentesque tristique bibendum libero sit amet lacinia. Aliquam sagittis eget risus cursus lobortis.

Submit

### Patch Function Code

```
Patch(  
  'Work Orders',  
  LookUp('Work Orders', ID=1),  
  {Notes: TextInput_Notes.Text}  
)
```

# PATCH A Number Column

## Goal

Change the Score column (Number type) from 55 to 80

## Input

Test Scores (SharePoint List)

ID	Student Name	Score
1	Samuel Wright	55

## Patch Function Code

```
Patch(  
  'Test Scores',  
  Lookup('Test Scores', ID=1),  
  {Score: 80}  
)
```

## Output

Test Scores (SharePoint List)

ID	Student Name	Score
1	Samuel Wright	80

## Scenario #1: Text Input

Patch a value found in a *text input* to a Number field.

Score

Submit

## Patch Function Code

```
Patch(  
  'Test Scores',  
  LookUp('Test Scores', ID=1),  
  {Score: Value(TextInput_Score.Text)}  
)
```

## Scenario #2: Dropdown

Patch a value found in a *dropdown* to a Number field.

Score

80	▼
55	
65	
75	
80	
90	
100	

Submit

## Patch Function Code

```
Patch(  
  'Test Scores',  
  LookUp('Test Scores', ID=1),  
  {Score: Dropdown_Score.Selected.Value}  
)
```

## Scenario #3: Slider

Patch a value found in a *slider* to a Number field.

Score

85  

## Patch Function Code

```
Patch(  
  'Test Scores',  
  LookUp('Test Scores', ID=1),  
  {Score: Slider_Score.Value}  
)
```

# PATCH A Yes/No Column

## Goal

Change the Completed column (Number type) from No to Yes

## Input

IT Helpdesk Tickets (SharePoint List)

ID	Date	Description	Completed
1	11/7/2020	Computer Not Turning On	No

## Patch Function Code

```
Patch(  
  'IT Helpdesk Tickets',  
  Lookup('IT Helpdesk Tickets', ID=1),  
  {Completed: true}  
)
```

## Output

IT Helpdesk Tickets (SharePoint List)

ID	Date	Description	Completed
1	11/7/2020	Computer Not Turning On	Yes

## Scenario #1: Toggle

Patch a value found in a *toggle* to a Yes/No field.

Completed



On



### Patch Function Code

```
Patch(  
  'IT Helpdesk Tickets',  
  Lookup('IT Helpdesk Tickets', ID=1),  
  {Completed: Toggle_Completed.Value}  
)
```

## Scenario #2: Dropdown

Patch a value found in a *dropdown* to a Yes/No field.

Completed

Yes	▼
No	
Yes	



## Patch Function Code

```
Patch(
  'IT Helpdesk Tickets',
  Lookup('IT Helpdesk Tickets', ID=1),
  {
    Completed: If(
      Dropdown_Completed.Value="No",
      false,
      Dropdown_Completed.Value="Yes",
      true
    )
  }
)
```



# PATCH A Person Column

## Goal

Change the EmployeePerson column (Person Type) from *blank* to “Matthew Devaney” in a SharePoint list.

## Input

Project Tasks (SharePoint List)

ID	TaskName	DueDate	EmployeePeople
1	Write project plan	11/7/2020	

## Patch Function Code

```
Patch(  
  'Project Tasks',  
  Lookup('Project Tasks', ID=1),  
  {  
    EmployeePerson: {  
      '@odata.type': "#Microsoft.Azure.Connectors.SharePoint.SPListExpandedUser",  
      Claims: "i:0#.f|membership|mdevaney@companyname.com",  
      Department: "",  
      DisplayName: "Matthew Devaney",  
      Email: "mdevaney@companyname.com",  
      JobTitle: "",  
      Picture: ""  
    }  
  }  
)
```

## Output

Project Tasks (SharePoint List)

ID	TaskName	DueDate	EmployeePeople
1	Write project plan	11/7/2020	Matthew Devaney

## Scenario #1: ComboBox

Patch a value found in a *combo box* to a People field.

Employee

Submit

### *ComboBox Properties Code*

```
Items: Choices('PATCH Function Examples'.EmployeePerson)
DisplayFields: ["DisplayName"]
SearchFields: ["DisplayName"]
```

### *Patch Function Code*

```
Patch(
  'Project Tasks',
  LookUp('Project Tasks', ID=1),
  {EmployeePerson: ComboBox_Person.Selected}
)
```

# PATCH A Person Column With Multiple People

## Goal

Change the EmployeePeople column (Person Type) from *blank* to “Matthew Devaney”, “Sarah Green” and “David Johnson” in a SharePoint list.

## Input

Project Tasks (SharePoint List)

ID	TaskName	DueDate	EmployeePeople
1	Write project plan	11/7/2020	

## Patch Function Code

```
Patch(
  'Project Tasks',
  Lookup('Project Tasks', ID=1),
  {
    EmployeePeople: Table(
      {
        '@odata.type': "#Microsoft.Azure.Connectors.SharePoint.SPListExpandedUser",
        Claims: "i:0#.f|membership|mdevaney@companyname.com",
        Department: "",
        DisplayName: "Matthew Devaney",
        Email: "mdevaney@companyname.com",
        JobTitle: "",
        Picture: ""
      },
      {
        '@odata.type': "#Microsoft.Azure.Connectors.SharePoint.SPListExpandedUser",
        Claims: "i:0#.f|membership|sgreen@companyname.com",
        Department: "",
        DisplayName: "Sarah Green",
        Email: "sgreen@companyname.com",
        JobTitle: "",
```

```
        Picture: ""
    },
    {
        '@odata.type': "#Microsoft.Azure.Connectors.SharePoint.SPListExpandedUser",
        Claims: "i:0#.f|membership|djohnson@companyname.com",
        Department: "Technology",
        DisplayName: "David Johnson",
        Email: "djohnson@companyname",
        JobTitle: "",
        Picture: ""
    }
)
}
```

### Output

Project Tasks (SharePoint List)

ID	TaskName	DueDate	EmployeePeople
1	Write project plan	11/7/2020	Matthew Devaney Sarah Green David Johnson

### Scenario #1: ComboBox

Patch a value found in a *combo box* to a People field.

Employee

## *ComboBox Properties Code*

```
Items: Choices('PATCH Function Examples'.EmployeePeople)
DisplayFields: ["DisplayName"]
SearchFields: ["DisplayName"]
```

## *Patch Function Code*

```
Patch(
  'Project Tasks',
  LookUp('Project Tasks', ID=1),
  {EmployeePeople: ComboBox_Person.SelectedItems}
)
```

# PATCH A Date Column

## Goal

Change the VacationDate column (Date Type) from 11/7/2020 to 12/25/2020.

## Input

Vacation Requests (SharePoint List)

ID	EmployeeName	VacationDate	Hours
1	Matthew Devaney	11/7/2020	8

## Patch Function Code

```
Patch(  
  'Vacation Requests',  
  Lookup('Vacation Requests', ID=1),  
  {VacationDate: Date(2020,12,25) }  
)
```

## Output

Vacation Requests (SharePoint List)

ID	EmployeeName	VacationDate	Hours
1	Matthew Devaney	12/25/2020	8

## Scenario #1: Date Picker

Patch a value found in a *date picker* to a Date field.

Vacation Date

### *Patch Function Code*

```
Patch(  
  'Vacation Requests',  
  LookUp('Vacation Requests', ID=1),  
  {VacationDate: DatePicker_VacationDate.SelectedDate}  
)
```

# PATCH A Date & Time Column

## Goal

Change the VacationStartDateTime column (Date & Time Type) from *blank* to *12/25/2020 9:30AM*.

## Input

Vacation Requests (SharePoint List)

ID	EmployeeName	VacationStartDateTime	Hours
1	Matthew Devaney		8

## Patch Function Code

Patch(

```
Patch(  
  'Vacation Requests',  
  Lookup('Vacation Requests', ID=1),  
  {VacationStartDateTime: Date(2020,12,25) + Time(9,30,0)}  
)
```

## Output

Vacation Requests (SharePoint List)

ID	EmployeeName	VacationStartDateTime	Hours
1	Matthew Devaney	12/25/2020 9:30AM	8



## Scenario #1: Date & Time Picker

Patch a value found in a *date & time picker* to a Date & Time field.

Vacation Date

  :   :  

### Patch Function Code

```
Patch(
  'Vacation Requests',
  LookUp('Vacation Requests', ID=1),
  {
    VacationStartDateTime:
      DatePicker_VacationDate.SelectedDate
      + Time(
        Value(Dropdown_Hours.Selected.Value),
        Value(Dropdown_Minutes.Selected.Value),
        0
      )
  }
)
```

# PATCH A Choices Column

## Goal

Change the StatusChoice column (Choices Type) from “Submitted” to “Approved” in a SharePoint List

## Input

Travel Expenses (SharePoint List)

ID	Item	Amount	StatusChoice
1	Steak Dinner	52.50	Submitted

## Patch Function Code

```
Patch(  
  'Travel Expenses',  
  Lookup('Travel Expenses', ID=1),  
  {StatusChoice: {Value: "Approved"}}  
)
```

## Output

Travel Expenses (SharePoint List)

ID	Item	Amount	Status
1	Steak Dinner	52.50	Approved

## Scenario #1: Combo Box

Patch a value found in a *combo box* to a Choices field.

Status



A screenshot of a web form. On the left, there is a label 'Status' above a dropdown menu. The dropdown menu is open, showing 'Approved' as the selected option. To the right of the dropdown is a blue button with the text 'Submit'.


### Patch Function Code

```
Patch(  
  'Travel Expenses',  
  LookUp('Travel Expenses', ID=1),  
  {Status: Combobox_Status.Selected}  
)
```

## Scenario #2: Dropdown

Patch a value found in a *dropdown* to a Choices field.

Status



A screenshot of a web form. On the left, there is a label 'Status' above a dropdown menu. The dropdown menu is open, showing a list of options: 'Approved', 'Submitted', 'Approved', and 'Rejected'. The second 'Approved' option is highlighted. To the right of the dropdown is a blue button with the text 'Submit'.

## Patch Function Code

```
Patch(  
  'Travel Expenses',  
  Lookup('Travel Expenses', ID=1),  
  {Status: Dropdown_Status.Selected}  
)
```

## Scenario #3: Dropdown

Patch a value found in *radio buttons* to a Choices field.

Status

- Submitted  
 Approved  
 Rejected

Submit

## Patch Function Code

```
Patch(  
  'Travel Expenses',  
  Lookup('Travel Expenses', ID=1),  
  {Status: Radio_Status.Selected.Value}  
)
```

# PATCH A Choices Column With Multiple Values

## Goal

Change the ColorsChoice column (Choices Type) from *blank* to the values “Blue”, “White” and “Red”

## Input

*Painting Jobs* (SharePoint List)

ID	Address	JobDate	ColorsChoice
1	30 State Street	11/1/2020	

## Patch Function Code

```
Patch(  
  'Painting Jobs',  
  Lookup('Painting Jobs', ID=1),  
  {  
    ColorsChoice: [  
      {Value: "Red"},  
      {Value: "White"},  
      {Value: "Blue"}  
    ]  
  }  
)
```

## Output

*Painting Jobs* (SharePoint List)

ID	Item	Amount	Status
1	Steak Dinner	52.50	Red White Blue

### Scenario #1: ComboBox

Patch a value found in a *combo box* to a Choices field.

Colors

A screenshot of a user interface. On the left, there is a multi-select dropdown menu with a blue border and a white background. The menu is open, showing three selected items: 'Red', 'White', and 'Blue', each in a light blue box. To the right of the menu is a blue button with a white downward-pointing chevron. To the right of the dropdown menu is a blue rounded rectangular button with the word 'Submit' in white text.

### Patch Function Code

```
Patch(  
  'Painting Jobs',  
  LookUp('Painting Jobs', ID=1),  
  {ColorsChoice: ComboBox_Colors.SelectedItems}  
)
```

# PATCH A LookUp Column

## Goal

Change the *BranchLocationLookUp* column (Lookup Type) from “Fargo, ND” to “Minneapolis, MN” in a SharePoint List

## Input

*Site Inspections* (SharePoint List)

ID	InspectionName	Date	BranchLocationLookUp
1	Quality Audit	11/7/2020	Fargo, ND

*Branch Locations* (SharePoint List)

ID	Branch Location
1	Fargo, ND
2	Minneapolis, MN

## Patch Function Code

```
Patch(  
  'Site Inspections',  
  Lookup('Site Inspections', ID=1),  
  {  
    BranchLocationLookUp: {  
      Id: 2,  
      Value: "Minneapolis, MN"  
    }  
  }  
)
```

## Alternate Code

```
Patch(
  'Site Inspections',
  Lookup('Site Inspections', ID=1),
  {
    BranchLocationLookup: {
      Id: 2,
      Value: Lookup('Branch Locations', ID=2, Title)
    }
  }
)
```

## Output

*Site Inspections* (SharePoint List)

ID	InspectionName	Date	BranchLocationLookup
1	Quality Audit	11/7/2020	Minneapolis, MN

## Scenario #1: Combo Box

Patch a value found in a *combo box* to a Lookup field.

Branch Location



## *Patch Function Code*

```
Patch(  
  'Site Inspections',  
  LookUp('Site Inspections', ID=1),  
  {BranchLocationLookUp: ComboBox_BranchLocation.Selected}  
)
```

# PATCH An Image Column

## Goal

Upload an image called *Business Cat* to the *AnimalImages* field.

## Input

*Animal Pictures* (SharePoint List)

ID	Description	AnimalImages
1	A picture of a cat	


## Patch Function Code

Uploads a file called *Business-Cat.jpg* to the *AnimalImages* field.

```
Patch(  
  'Travel Expenses',  
  Lookup('Travel Expenses', ID=1),  
  {AnimalImages: 'Business-Cat'}  
)
```

## Output

*Animal Pictures* (SharePoint List)

ID	Description	AnimalImages
1	A picture of a cat	

## Scenario #1: Add Picture

Patch a value found in an *Add Picture* control to an Image field.

]



### *Patch Function Code*

```
Patch(  
  'Travel Expenses',  
  LookUp('Travel Expenses', ID=1),  
  {AnimalImage: AddMediaButton1.Media}  
)
```

## Scenario #2: Pen Input

Patch a signature found in a *Pen Input* control to an Image field.



## Patch Function Code

```
Patch(  
  'Travel Expenses',  
  LookUp('Travel Expenses', ID=1),  
  {AnimalImage: PenInput1.Image}  
)
```

## Scenario #3: Camera

Patch a photo found in a *Camera* control to an Image field.



## Patch Function Code

Note: the user must first click on the Camera control to load the Photo property with an image.

```
Patch(  
  'Travel Expenses',  
  LookUp('Travel Expenses', ID=1),  
  {AnimalImage: Camera1.Photo}  
)
```

# PATCH A Hyperlink Column

## Goal

Change the WebsiteHyperlink column (Hyperlink Type) from *blank* to "https://www.google.com".

*Special Note:* a hyperlink column has two properties – Url and Description. Power Apps patches the Url and Description with the same value. If you want to have different values for these you must use Microsoft Automate as a [workaround](#).

## Input

Vendors (SharePoint List)

ID	VendorName	WebsiteHyperlink
1	Matthew Devaney	

## Patch Function Code

```
Patch(  
  Vendors,  
  Lookup(Vendors, ID=1),  
  {WebsiteHyperlink: "https://www.google.com"}  
)
```

## Output

Vendors (SharePoint List)

ID	VendorName	WebsiteHyperlink
1	Matthew Devaney	https://www.google.com

## Scenario #1: Text Input

Patch a value found in a *text box* to a Hyperlink field.

Website

### *Patch Function Code*

```
Patch(  
  Vendors,  
  LookUp(Vendors, ID=1),  
  {WebsiteHyperlink: TextInput_Website.Text}  
)
```

# PATCH A Currency Column

## Goal

Change the TotalAmountCurrency column (Currency type) from *blank* to \$350.25

## Input

Customer Invoices (SharePoint List)

ID	InvoiceNumber	CustomerName	TotalAmountCurrency
1	1001	HYK Group Limited	

## Patch Function Code

```
Patch(  
  'Customer Invoices',  
  Lookup('Customer Invoices', ID=1),  
  {TotalAmountCurrency: 350.25}  
)
```

## Output

Customer Invoices (SharePoint List)

ID	InvoiceNumber	CustomerName	TotalAmountCurrency
1	1001	HYK Group Limited	\$320.25

## Scenario #1: Text Input

Patch a value found in a *combo box* to a Currency field.

Total Amount

### *Patch Function Code*

```
Patch(  
  'Customer Invoices',  
  LookUp('Customer Invoices', ID=1),  
  {TotalAmountCurrency: TextInput_TotalAmount.Text}  
)
```



# PATCH A Task Outcome Column

## Goal

Change the StatusTaskOutcome column (Choices Type) from *blank* to “Approved” in a SharePoint List

## Input

Travel Expenses (SharePoint List)

ID	Item	Amount	StatusTaskOutcome
1	Steak Dinner	52.50	

## Patch Function Code

```
Patch(  
  'Travel Expenses',  
  Lookup('Travel Expenses', ID=1),  
  {StatusTaskOutcome: {Value: "Approved"}}  
)
```

## Output

Travel Expenses (SharePoint List)

ID	Item	Amount	Status
1	Steak Dinner	52.50	Approved

## Scenario #1: Combo Box

Patch a value found in a *combo box* to a task outcome field.

Status



A screenshot of a web form. On the left, there is a dropdown menu labeled 'Status' with 'Approved' selected. To the right of the dropdown is a blue button with the text 'Submit'.


### Patch Function Code

```
Patch(  
  'Travel Expenses',  
  LookUp('Travel Expenses', ID=1),  
  {Status: Combobox_Status.Selected}  
)
```

## Scenario #2: Dropdown

Patch a value found in a *dropdown* to a task outcome field.

Status



A screenshot of a web form. On the left, there is a dropdown menu labeled 'Status' with 'Approved' selected. The dropdown menu is open, showing a list of options: 'Submitted', 'Approved' (highlighted), and 'Rejected'. To the right of the dropdown is a blue button with the text 'Submit'.

## Patch Function Code

```
Patch(  
  'Travel Expenses',  
  LookUp('Travel Expenses', ID=1),  
  {Status: Dropdown_Status.Selected}  
)
```

## Scenario #3: Dropdown

Patch a value found in *radio buttons* to a task outcome field.

Status

- Submitted
- Approved
- Rejected

Submit

## Patch Function Code

```
Patch(  
  'Travel Expenses',  
  LookUp('Travel Expenses', ID=1),  
  {Status: Radio_Status.Selected.Value}  
)
```